
1 Introduction

The Naproche system (Natural Proof Checking) accepts input texts in the artifical lan-
guage ForTheL (Formula Theory Language) which approximates the ordinary language
of mathematics. The language of mathematics combines natural language with symbolic
terms. ForTheL uses a minimal subset of English as its natural language which can be
extended interactively by mathematical notions and phrases.

The Naproche system goes back to the System for Automated Deduction (SAD) by Andrei
Paskevich: Méthodes de formalisation des connaissances et des raisonnements mathéma-
tiques: aspects appliqués et théoriques. PhD thesis, 2007. Information and a web interface
to SAD is available at http://nevidal.org/sad.en.html. In recent years, the Naproche
group at Bonn has taken over and extended the original code of SAD and the ForTheL
language, including a LATEX input format. Originally a command line tool, Naproche can
now be also used within the Isabelle Proof Interactive Development Environment (PIDE).

This document is a two-part introduction to Naproche. The first part presents general
principles. The second half is written around a version of Euclid's well-known proof of
the infinitude of primes. The Euclid example is also available in the Isabelle/Naproche
examples folder as euclid.. and as TUTORIAL.. The introduction is in a preliminary state,
and the software is under constant development. Comments for improvement are welcome
(koepke@math.uni-bonn.de).

1.1 Running ℕaproche in Isabelle

Naproche is available as a component of the Isabelle prover environment. Isabelle can be
downloaded from https://isabelle.in.tum.de/. The Linux and macOS versions have
to be unpacked in some convenient folder. Isabelle can be started by issuing the command
isabelle or ./isabelle in that folder. The Windows version is a self-installing exe and
should register Isabelle in the start menu. Isabelle is a big package whose download,
installation and start-up will take several minutes.

The main Isabelle window provides an editor buffer together with output und file browser
buffers. Editing files with a .ftl or .ftl.tex file extension in the Isabelle editor will
activate continuous checking of the editor buffer by Naproche.

Naproche feedback is available through the output buffer, or by hovering the mouse over
the input, or by colored text highlighting.

Note that the checking process consists of proving a large number of explicit and implicit
first-order properties of the ForTheL text by the automatic theorem prover E. The power
of E and of the overall Naproche system strongly depends on hardware performance. It
may be necessary to supply more proof details to get a text checked. Texts may well take
several minutes for checking.

1.2 Natural Language Processing

The ForTheL input text is interpreted in the target logic and it also proposes proof methods
to be used by the reasoner and the ATP. ForTheL leverages a number of natural language
mechanisms to capture formal content in a compact, user-friendly and natural way. This
corresponds to usual natural language features, where the phrase �white horse that belongs
to Mary� with its adjective, noun and relative sentence corresponds to a first-order state-
ment like

horse(x)^white(x)^ property¡ of (x;Mary)

with a (hidden) variable x, predicates horse(), white(), and property¡of(,), and a constant
Mary.

The natural language parser of the Naproche system extracts this formal content whilst
reading the input sentence by sentence. Previous sentences provide the context of already
introduced language components, in which the new sentence is to be interpreted. Similar
comments apply to the symbolic content of texts. Indeed Naproche allows to mix natural
language and symbolic phrases, as does the ordinary language of mathematics.

1.3 Axiomatic Approach
The Naproche system comes with a minimal set of in-built mathematical notions. Usually
one has to explicitly extend the first-order language through Signature and Definition
commands and through Axioms. Then Lemmas and Theorems can be postulated and
proved with familiar proof structures.

1.4 Defining Notions
In mathematical English, words often attain meanings different from their ordinary usage
and the same word may even have different meanings in different areas of mathematics.
A �natural number� is a specific mathematical notion which is introduced and defined in
the language of arithmetic. Its meaning may have some vague connection to the ordinary
English word �natural�, but this is not necessarily so. A �natural transformation� in cate-
gory theory involves a completely different usage of the word �natural�.

Similarly symbols may have different meanings in different contexts. �0� may denote the
natural number 0, or a zero matrix, or a certain �neutral element� in various structures.

The �vocabulary� of Naproche can be extended by Signature and Definition commands
to introduce notions and constants. Natural numbers and a specific natural number 0 can
be introduced by putting the following commands in a .ftl file:

Signature. A natural number is a notion.

Signature. 0 is a natural number.

The effects of the introduction of natural numbers and of 0 can be viewed as delineating
a subdomain N in a �universe� of all mathematical objects and as picking a specific object
in the subdomain.

N

0

1

���

R

0

���

���

(G; �)

G1
G2

We thus imagine a universe of mathematical objects of which we cut out certain domains
for closer investigation: Besides the standard structures like N and R the picture suggests
several groups G; G1; G2; ::: , which themselves could be collected together into, e.g., a
category of groups (dotted line). And we could collect together all categories to study the
general notion of a category (blue line).

The internal effect of these commands can be inspected in the output window:

[Translation] (line 5 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
forall v0 ((HeadTerm :: aNaturalNumber(v0)) implies truth)
[Translation] (line 9 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
forall v0 ((HeadTerm :: v0 = 0) implies aNaturalNumber(v0))

So a unary internal property aNaturalNumber(.) and a constant symbol 0 have been
introduced together with the type information that 0 is a natural number. This is expressed
by the second translation; leaving away some system orientated tagging this is basically
the logic formula:

8v0(v0=0! aNaturalNumber(v0)):

Naproche expects words and symbols to have unique meanings. If we would also declare:

Signature. A matrix is a notion.

Signature. 0 is a matrix.

then the symbol 0 is �overloaded� and its use is ambiguous. A statement like 0=0 would
now be rejected by Naproche with an ambiguity error.

1.5 Input Formats

There are two formats of the ForTheL language: a classic one using standard ASCII and
some ASCII-art for symbolic material, indicated by a .ftl file extension; and a LaTeX-
based version with a .ftl.tex extension. In this note we shall mainly use the latter version,
with the possibility of immediate mathematical typesetting by LATEX.

The .ftl file

Signature. A natural number is a notion.

Signature. 0 is a natural number.

has a �pretty-printed� LATEX version:

Signature. A natural number is a notion.

Signature. 0 is a natural number.

which corresponds to a source xyz.ftl.tex:

\begin{document}

\begin{forthel}

\begin{signature}

A natural number is a notion.

\end{signature}

\begin{signature}

$0$ is a natural number.

\end{signature}

\end{forthel}

\end{document}

Naproche provides a LATEX environment

\begin{signature} . . . \end{signature}

and similar environments for definitions, theorems, . . . :

\begin{definition} . . . \end{definition}
\begin{theorem} . .. \end{theorem}
\begin{proof} . . . \end{proof}

Only content in

\begin{forthel} ... \end{forthel}

environments is passed to Naproche, and one can write arbitrary material outside those
environments. Documents should thus be structured like

\begin{document}

. . .

[Arbitrary text.]

\begin{forthel}

. . .

[ForTheL commands and statements]

. . .

\begin{forthel}

. . .

\begin{forthel}

. . .

\end{forthel}

. . .

\end{document}

ForTheL files with a proper LATEX preamble and a \begin{document} ... \end{document}
structuring can be immediately compiled by TEX/LATEX once the forthel environment is
defined in some package or style file.

To experiment with texts in .ftl.tex format one can edit the forthel environments. For
this it may be convenient to deactivate most of these environments by replacing the outer
\begin and \end by, e.g., begin and end.

1.5.1 Natural Language Aspects of Signature Commands

When Naproche encounters the two commands

Signature. A natural number is a notion.

Signature. 0 is a natural number.

the system first detects the kind of command given, and then looks for patterns of words
and symbols in particular positions of the command.

The first command is of the form

Signature. A ... is a notion.

This command expects a new pattern of words and symbols in the . . . position, and a
word is simply a sequence of alphabetical symbols. For the software, the particular choice
of words is completely irrelevant as long that it leads to well-distinguished identifiers for
further use. So could have used a �Hilbertian variant� like

Signature. A Bierseidel is a notion.

Signature. 0 is a Bierseidel.

Or wild letter combinations like AbcDE. The choice of words is a question of mathematical
style and is the responsibility of the user of the system.

There are many grammatical simplifications in the system: articles like �a� and �the� are
sometimes (but not always, see above) ignored. Plural forms are treated as singular, etc.
Words in patterns are internally transformed to all lowercase, and again the correct choice
in the text is a question of grammatical correctness. To get a feel for such identifications,
one should experiment with texts and look at their translations and behaviour.

Grammatical flexions of various kinds can be used via synonyms which are introduced by
synonym commands to the parser. Writing

[synonym number/numbers]

identifies the two words �number� and �numbers� so that the correct form singular or plural
form of �natural number� can be inserted where needed. To use this for the pattern �natural
number�, the synonym command has to appear before the pattern introduction.

One can also declare aliases for all sorts of terms by commands of the form

Let an . . . stand for . . . .

. We could, e.g., use �integer� instead of �natural number� as follows:

[synonym number/numbers]

Signature. A natural number is a notion.

Let an integer stand for natural numbers.

Signature. 0 is a natural number.

The symbolic pattern in the last command can be much more complex than just 0 . We
could for example use the constant �~ ($\tilde{\sigma}$):

Signature. �~ is an integer.

6



Note that this command for the introduction of a constant is of the form

Signature. . . . is [an] . . . .

This command is slightly different from

Signature. The .. . is [an] . . . .

If one writes

Signature. Zero is an integer.

then this introduces a symbolic constant that can only be written as �Zero� and not as
�zero�. On the other hand

Signature. The zero is an integer.

introduces a word (pattern) �zero� that can also be written as �Zero�.

The treatment of natural language features in ForTheL and Naproche is at the same time
crude and subtle. It is modeled after the mathematical literature and should give good
results on average. To get a better feeling, one should experiment with the system, or, even
better, understand the behaviour by reading the code.

Exercise 1. Introduce constants ONE, TWO, THREE to the natural numbers, that have to be
written exactly in this form. Alternatively introduce a word �one� which can also be capitalized at the
beginning of a sentence.

Exercise 2. Introduce vector spaces by fixing a domain of scalars and a domain of vectors. Moreover
introduce a scalar 0 and a zero vector 0~ . Provide aliases so that the 0~ can also be called the �zero vector�.
How does Naproche react if one wants to prove an equality or inequality between a scalar and a vector?

Exercise 3. Provide a language adequate for the Boolean algebra f>;?g of truth values >=TRUE
and ?=FALSE .

1.6 Equality and Inequality Statements

ForTheL provides the symbols = (equality) and =/ (inequality). This allows to form state-
ments like t1= t2 and t1=/ t2 , where t1; t2 are terms �known� to Naproche. Let us start
setting up the natural numbers with constants 0; 1; 2 :

Signature. A natural number is a notion.

Signature. 0 is a natural number.

Signature. 1 is a natural number.

Signature. 2 is a natural number.

Some basic knowledge about equalities is available in the proving mechanisms ofNaproche.

So we formulate our first theorems and try:

Theorem 1. 0=0 .

Theorem 2. 0=/ 0 .

Theorem 3. not 0=/ 0 .

Theorem 4. 0=1 :

Theorem 5. 0=/ 1 .

In the classic .ftl format these theorems can be written as

Theorem 1. 0 = 0.
Theorem 2. 0 != 0.
Theorem 3. not 0 != 0.
Theorem 4. 0 = 1.
Theorem 5. 0 != 1.

The theorems translate to:

0= 0
not 0= 0

not not 0=0
0=1

not 0= 1

They are given to an automated theorem prover (ATP) as conjectures that have to be
proven from current assumptions. One can see these �prover tasks� by putting the command
[dump on] before the theorem to be proved. In case of Theorem 2 the output buffer
contains:

[Translation] (line 27 of "/home/koepke/TEST1/Temp/temp.ftl.tex")
0 = 0
[Reasoner] (line 27 of "/home/koepke/TEST1/Temp/temp.ftl.tex")
goal: 0 = 0 .
[Main] (line 27 of "/home/koepke/TEST1/Temp/temp.ftl.tex")
fof(m_,hypothesis,$true).
fof(m_,hypothesis,aElement(sz0)).
fof(m_,hypothesis,aElement(szi)).
fof(m__,conjecture,(sz0 = sz0)).

So the parser which translates the input to formal logic translates to 0 = 0.

The reasoner identifies prover tasks, i.e., goals to be proved: 0 = 0.

Then the main Naproche program sends the four bottom lines to the automated theorem
prover in a special format; fof stands for �first-order formula�; lines with hypothesis have
been assumed or proved earlier like aElement(sz0) from the signature introduction of the
constant 0; note that identifiers are sometimes modified for programming reasons; lines
with conjecture have to be proved by the ATP. The ATP can signal success or failure
which will then be signaled to the user in various ways.

Theorem 2 is checked positively, because the reflexivity of = is an in-built assumption.
Therefore Theorem 3 is checked negatively: the search for a proof of the theorem fails.

Actually, in case of the �false� Theorem 3, the reasoner makes two attempts which both fail:

fof(m_,hypothesis,$true).
fof(m_,hypothesis,aElement(sz0)).
fof(m_,hypothesis,aElement(szi)).
fof(m__,conjecture,( ~ (sz0 = sz0))).

[Main] (line 27 of "/home/koepke/TEST1/Temp/temp.ftl.tex")
fof(m_,hypothesis,$true).
fof(m_,hypothesis,aElement(sz0)).
fof(m_,hypothesis,aElement(szi)).
fof(m__,conjecture,( ~ (aElement(sz0) & (aElement(sz0) & (aElement(sz0) &
(aElement(sz0) & (sz0 = sz0))))))).

In the second round the reasoner tries, again without success, to �help� the prover by adding
some information directly to the claim.

Theorem 2 must fail, because in Theorem 3 we successfully prove the opposite.

Remark 6. Naproche first parses a text and translates it into its internal logic. If this
step is completed, it starts to logically check the (internal) text until the first failure. So
if one wants to check more of the text, one has to exclude initial failures by inserting \
end{forthel} and \begin{forthel} into the text.

Concerning Theorems 5 and 6, both have to fail, because the system has no information
about the equality or inequality of 0 and 1. In certain circumstances they may agree, in
others not. Note that an identifiers like 1 may suggest to human readers that this is the
standard number 1 which in the standard model is =/0. But this implicit information is
not available to Naproche, and indeed in axiomatizations of algebraic structures one often
requires explicitely that 0=/ 1.

We can also form theorems with assumptions of the form �Assume that . . . � where . . . is
a statement:

Theorem. Assume that 0= 1. Then 1=0.

Theorem. Assume that 0= 1. Assume that 1=2 . Then 0= 2 .

Exercise 4. State and prove a theorem that =/ is symmetric.

1.6.1 Natural Language Aspects of Theorems

The Theorem command of ForTheL is invoked by the theorem environment of LATEX:

\begin{theorem}

. . .

\end{theorem}

9



There are several linguistic degrees of freedom that capture common variants in the liter-
ature.

One can alternatively use the LATEX lemma, corollary, proposition environments.

A theorem begins with zero or finitely many assumptions, which are of the form

([let us | we can] (assume | presume | suppose) [that] | let)
. . .

Alternatives are indicated by |, options by [ . . . ]; . . . stands for a statement.

After the assumptions the claim of the theorem consists of exactly one statement which is
optionally prefixed by

[then | hence | thus | therefore | consequently]

Optional words can be viewed as filler words that make for better reading but are logically
redundant. Before symbolic statements of the form t0= t1 or t0=/ t1 one may fill in the
phrase �we have�:

Theorem. Assume that we have 0=/ 1. Then we have 0=/ 2 or we have 1=/ 2 .

Exercise 5. In the language of natural numbers introduced so far one can express that the constants
1; 2; :::; n are pairwise unequal by n � (n¡ 1)

2
assumptions of the form

Assume that i=/ j.

where 16 i< j6n . Can this also be expressed by less than n � (n¡ 1)
2

of those inequality assumptions?

ForTheL provides a shorthand for these inequalities: the statement

0,1,2,3 are pairwise nonequal.

translates to

[Translation] (line 29 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
(((((not 0 = 1 and not 0 = 2) and not 0 = 3) and not 1 = 2) and not 1 =
3) and not 2 = 3)

If we omit the word �pairwise� we get

0,1,2,3 are nonequal.

and

[Translation] (line 29 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
((not 0 = 1 and not 1 = 2) and not 2 = 3)

which means that (only) consecutive members of 0; 1; 2; 3 are unequal.

1.6.2 Natural Language Connectives and Boolean Operations

Statements ' and  can be combined to complex statements. In ForTheL, standard natural
language connectives are available, with some natural language alternatives:

¡ not ', alternatively: it is false that ';

¡ ' or  ;

¡ ' and  ;

¡ if ' then  ;

¡ ' iff  , alternatively: ' if and only if  .

Boolean operations

We have proved that 0=0 is true and that 0=/ 0 is false. We can take these statements as
representatives for the truth values true and false. The alias mechanism can be applied to
complete statements:

Let TRUE stand for 0= 0 .
Let FALSE stand for 0=/ 0 .

Applying these connectives to the truth values TRUE and FALSE corresponds to Boolean
operations in a 2-element Boolean algebra:

Exercise 6. Form combinations of the statements TRUE and FALSE with the above connectives and
and prove or disprove (i.e., prove the negation of) the combined statements. E.g., try to prove

Theorem. TRUE or FALSE.

or

Theorem. Not (TRUE or FALSE).

This exercise can be interpreted as showing that TRUE and FALSE form a Boolean algebra
with the operations �and�, �or�, and �not�.

1.7 Variables

Variables like x; y; z; ::: are characteristic for mathematical statements, and they are crucial
for the strength of such statements: a variable stands for all possible elements of some
notion so that statements with variables may include an infinity of cases.

In ForTheL, variables are typed as elements of notions. They can be introduced by �pre-
typing commands� like:

Let x; y; z denote natural numbers.
Let �; �;  stand for integers.

This is a declaration, which introduces the identifiers �x�, �y� and �z� as variables that range
over natural numbers. If x is found in a later statement, the typing by natural numbers is
automatically recalled and used for the processing of x.

After the above declarations we can formulate and prove theorems like

Theorem. x=x.

Theorem. If x= y then y=x .

Theorem. If x= y and y= z then x= y .

These theorems express that the relation = is reflexive, symmetric and transitive, i.e., that
it is an equivalence relation.

The translations of these theorems show that the type declarations of the occuring vari-
ables, and how these are given to the external prover. In case of transitivity:

[Translation] (line 25 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
((aNaturalNumber(x) and aNaturalNumber(y)) and aNaturalNumber(z))
[Translation] (line 25 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
((x = y and y = z) implies x = z)
[Reasoner] (line 25 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
goal: If x = y and y = z then x = z .
[Main] (line 25 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
fof(m_,hypothesis,$true).
fof(m_,hypothesis,( ! [W0] : (aNaturalNumber(W0) => (W0 = W0)))).
fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) &
aNaturalNumber(W1)) => ((W0 = W1) => (W1 = W0)))))).
fof(m__,hypothesis,((aNaturalNumber(xx) & aNaturalNumber(xy)) &
aNaturalNumber(xz))).
fof(m__,conjecture,(((xx = xy) & (xy = xz)) => (xx = xz))).

The translation also demonstrates that �free� variables in a theorem are interpreted as
freely moving over their type. The already established theorem on the symmetry of = now
appears as a hypothesis for the transitivity theorem:

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) &
aNaturalNumber(W1)) => ((W0 = W1) => (W1 = W0)))))).

The exclamation marks �!� are the universal quantifiers of the TPTP language for auto-
mated theorem proving. A more intuitive reading is the universally quantified formula:

8w08w1((aNaturalNumber(w0)^ aNaturalNumber(w1))! (w0=w1!w1=w0)):

All variables need to be declared through pretyping (or other means of declaration). If a
variable comes up without a declared type, the system will throw a parser error:

Theorem. u=u.

leads to the error:

[Parser] (line 29 of "/home/koepke/TEST1/Temp/temp02.ftl.tex")
(line 29, column 2):
free undeclared variables: u
in translation: u = u

Propositional logic

The truth value of a statement like 0=1 is so far not decided, but nevertheless we consider
it a properly formed mathematical statement and call it a proposition. Propositions can
also be combined into larger propositions by the above connectives �and�, �or�, etc.

Let us form properties PHI and PSI:

Let PHI stand for 0= 1.

Let PSI stand for 0=2.

Exercise 7. Study which complex statements are provable by Naproche. E.g.,

Theorem. If PHI then PHI.

Theorem. If PHI and if PHI then PSI then PSI.

The latter theorem is a basic rule of proving and is called modus ponens .

2 Formalization Example: Euclid's Proof

A first-order language is determined by its (symbols for) relations, functions, and constants.
We want to introduce the functions + and � of addition and multiplication (of natural
numbers) and the constants 0 and 1. We shall later consider other domains as well, like sets
and functions. The arithmetic functions and quantifiers will be restricted to the extension
of the unary relation symbol for natural numbers. The (weak) type system of ordinary
mathematical language is modeled by a system of first-order predicates. These types do
not follow a strict �type theory� with specific mathematical laws but they are still powerful
enough to organize the universe of mathematics.

In the following we demonstrate Naproche along a standard proof of the infinitude of
prime numbers:

� set up a language and axioms for natural number arithmetic;

� define divisibility and prime natural numbers;Bonn - Nordrhein-Westfalen

� introduce some set theory so that one can define finite sets, sequences and products.

Finally, a checked natural language proof of Euclid's theorem can be carried out in this
axiomatic setup.

Here is ForTheL code for introducing the type, or rather notion, of natural numbers, the
constants 0 and 1 and the operations of + and �.

[synonym number/-s]

Signature 7. A natural number is a notion.

Let i; k; l;m; n; p; q; r denote natural numbers.

Signature 8. 0 is a natural number.

Let x is nonzero stand for x=/ 0.

Note that some natural language processing is taking place here: �nonzero� is introduced
within the phrase �x is nonzero� in an adjective position. So �nonzero� can be used as an
adjective which modifies �natural number�, like in:

Signature 9. 1 is a nonzero natural number.

2.1 Functions

We now introduce the usual arithmetic operations for natural numbers.

Signature 10. m+n is a natural number.

Signature 11. m �n is a natural number.

These commands introduce new function symbols to the language. We can find them in the

first-order translation:

4. (aNaturalNumber(m) and aNaturalNumber(n))

5. forall v0 ((HeadTerm :: v0 = m+n) implies aNaturalNumber(v0))

6. (aNaturalNumber(m) and aNaturalNumber(n))

7. forall v0 ((HeadTerm :: v0 = m*n) implies aNaturalNumber(v0))

So addition + and multiplication � are now function symbols of the internal logical lan-
guage. In the Signature commands these are declared as symbolic patterns of the form _ +
_ and _ * _ ; these patterns are recognized since the parser is first looking for declared
variables in the signature statement. These determine the �holes� in the sequence of words
and symbols, and thus the newly agreed pattern. The number of holes determines the arity
of the function. The types of the variables in the holes determines the argument type of
the function, the �is a . . . � determines the value type. Thus we get the function type of +:

+: aNaturalNumber� aNaturalNumber! aNaturalNumber

Note that 5 and 7 both have the premises

(aNaturalNumber(m) and aNaturalNumber(n))

for the two arguments of the operations. In the process of �ontological checking� these
premises have to be proved before the operations can reasonably be applied within proofs.

2.2 Natural Numbers - Postulating Axioms

We now have to introduce axioms for our abstract first-order structure. Axiom are ForTheL
statements written in axiom environments.

Axiom 12. m+n=n+m.

Axiom 13. (m+n)+ l=m+(n+ l).

Axiom 14. m+0=m=0+m.

Axiom 15. m �n=n �m.

Axiom 16. (m �n) � l=m � (n � l).

Axiom 17. m � 1=m=1 �m.

Axiom 18. m � 0=0=0 �m.

Axiom 19. m � (n+ l)= (m �n)+ (m� l) and (n+ l) �m=(n �m)+ (l �m).

Axiom 20. If l+m= l+n or m+ l=n+ l then m=n.

Axiom 21. Assume that l is nonzero. If l �m= l �n or m � l=n � l then m=n.

Axiom 22. If m+n=0 then m=0 and n=0.

Axioms - like Signatures - are toplevel sections which consist of n+ 1 statements. The
first n are assumption statements (�Assume ...�, �Let ...�) under which the final statement
is postulated. Note that previous pretypings of variables also act like assumptions.

2.3 The Natural Order - Defining Relations and Functions

Definitions extend the first-order language by defined symbols as in the following examples
concerning the ordering of the natural numbers. A definition corresponds to a Signature
command in which a symbol is introduced plus an Axiom containing the defining property.

Definition 23. m�n iff there exists a natural number l such that m+ l=n.

Let m<n stand for m�n and m=/ n.

Definition 24. Assume that n�m. m¡n is a natural number l such that n+ l=m.

The first definition defines the binary relation � by an �iff� equivalence. This is followed
by a purely syntactic definition of <. m<n is an abbreviation for another formula. The
abbreviation will be expanded, possibly recursively, by the parser. The third definition
defines the binary difference function ¡.

Remark 25. Definitions of functions and constants usually contain implicit postulates
corresponding to the existence and uniquess-properties of function values and constants.
In case of the definition of ¡ the condition for l should be satisfiable by a unique natural
number. This is however not checked by Naproche, so that the well-definedness of the
function is the user's responsibility. If the function definition were non-unique we would
have a contradictory system of assumptions. Consider, e.g., the wrong definition

Definition 26. Assume that n�m. m¡n is a natural number l such that n=m.

The first-order translation would be

(aNaturalNumber(m) and aNaturalNumber(n))

n\leq m

forall v0 ((HeadTerm :: v0 = m-n)

iff (aNaturalNumber(v0) and n = m))

Every number fits the defining equivalence provided that m= n. But then 0= 0¡ 0= 1,
contradiction.

For relation definitions, these problems do not arise.

2.4 Lemmas and Theorems

After setting up the axiomatics we proceed to claim and prove properties. Claims together
with the accumulated axioms will be given to the background ATP (= eprover). Many
basic propositions can be proved by the ATP without further intervention. The following
three lemmas show that � is a partial order:

Lemma 27. m�m.

Lemma 28. If m�n�m then m=n.

Lemma 29. If m�n� l then m� l.

2.5 Eprover in the Background

These lemmas were checked correct by Naproche without explicit proofs. We can look
at the task given to the ATP by putting a [dump on] command in the beginning of the
ForTheL parts of the document and looking for the dump of the provertask in the output
window. The task is written in the first-order logic language TPTP which is a standard
input language for ATPs. Observe that all previous Signature, Axiom and Definition envi-
ronments can be found as premises of the conjecture m�m.

[Translation] (line 409 of ...

aNaturalNumber(m)

[Translation] (line 409 of ...

m\leq m

[Reasoner] (line 409 of ...

goal: m \leq m .

[Main] (line 409 of ...

fof(m_,hypothesis,$true).

fof(m_,hypothesis,aNaturalNumber(sz0)).

fof(m_,hypothesis,(aNaturalNumber(sz1) & ( ~ (sz1 = sz0)))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> aNaturalNumber(sdtpldt(W0,W1)))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> aNaturalNumber(sdtasdt(W0,W1)))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> (sdtpldt(W0,W1) = sdtpldt(W1,W0)))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ( ! [W2] : (((aNaturalNumber(W0) &

aNaturalNumber(W1)) & aNaturalNumber(W2))

=> (sdtpldt(sdtpldt(W1,W2),W0) = sdtpldt(W1,sdtpldt(W2,W0)))))))).

fof(m_,hypothesis,( ! [W0] : (aNaturalNumber(W0)

=> ((sdtpldt(W0,sz0) = W0) & (W0 = sdtpldt(sz0,W0)))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> (sdtasdt(W0,W1) = sdtasdt(W1,W0)))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ( ! [W2] : (((aNaturalNumber(W0) &

aNaturalNumber(W1)) & aNaturalNumber(W2))

=> (sdtasdt(sdtasdt(W1,W2),W0) = sdtasdt(W1,sdtasdt(W2,W0)))))))).

fof(m_,hypothesis,( ! [W0] : (aNaturalNumber(W0)

=> ((sdtasdt(W0,sz1) = W0) & (W0 = sdtasdt(sz1,W0)))))).

fof(m_,hypothesis,( ! [W0] : (aNaturalNumber(W0)

=> ((sdtasdt(W0,sz0) = sz0) & (sz0 = sdtasdt(sz0,W0)))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ( ! [W2] : (((aNaturalNumber(W0)

& aNaturalNumber(W1)) & aNaturalNumber(W2))

=> ((sdtasdt(W1,sdtpldt(W2,W0)) = sdtpldt(sdtasdt(W1,W2),sdtasdt(W1,W0)))

& (sdtasdt(sdtpldt(W2,W0),W1) = sdtpldt(sdtasdt(W2,W1),sdtasdt(W0,W1))))))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ( ! [W2] : (((aNaturalNumber(W0)

& aNaturalNumber(W1)) & aNaturalNumber(W2))

=> (((sdtpldt(W0,W1) = sdtpldt(W0,W2))

| (sdtpldt(W1,W0) = sdtpldt(W2,W0))) => (W1 = W2))))))).

fof(m_,hypothesis,( ! [W0] : (aNaturalNumber(W0) => (( ~ (W0 = sz0))

=> ( ! [W1] : ( ! [W2] : ((aNaturalNumber(W1) & aNaturalNumber(W2))

=> (((sdtasdt(W0,W1) = sdtasdt(W0,W2))

| (sdtasdt(W1,W0) = sdtasdt(W2,W0))) => (W1 = W2))))))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> ((sdtpldt(W0,W1) = sz0) => ((W0 = sz0) & (W1 = sz0))))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> (sdtbszlzezqdt(W0,W1)

<=> ( ? [W2] : (aNaturalNumber(W2) & (sdtpldt(W0,W2) = W1)))))))).

fof(m_,hypothesis,( ! [W0] : ( ! [W1] : ((aNaturalNumber(W0) & aNaturalNumber(W1))

=> (sdtbszlzezqdt(W1,W0) => ((aNaturalNumber(sdtmndt(W0,W1))

& (sdtpldt(W1,sdtmndt(W0,W1)) = W0)) & ( ! [W2] : ((aNaturalNumber(W2)

& (sdtpldt(W1,W2) = W0)) => (W2 = sdtmndt(W0,W1)))))))))).

fof(m__,hypothesis,aNaturalNumber(xm)).

2.6 Testing for Contradictions

It is quite common to accidentally introduce trivial inconsistencies in formalizations. Not
just by function definitions, but also because some marginal cases outside the main argu-
ment have not been treated right. E.g., although the number 0 is quite uninteresting for
the study of prime numbers, we still have to deal with 0-cases or explicitly request that
terms are nonzero. If a text with non-trivial mathematical content checks unexpectedly
fast then one should become suspicious.

To find inconsistencies it is helpful to try to prove

Lemma 30. Contradiction.

in various places of a text. If the lemma is validated byNaproche then one has to investigate
further. In the source text of this document one finds an Contradiction-Lemma which is
commented out by %. This can be quickly activated for a contradiction check. It can also
be used to force rechecking of the text: uncomment the lemma and then comment it again;
this will lead to rechecking at least from the position of the lemma onwards.

2.7 Linear and Discrete Orders

We need more axiomatic assumptions for the ordering of the natural numbers. The axioms
so far do not guarantee that the ordering is linear. Also we want a �discrete� order with
nothing strictly between 0 and 1. So we continue:

Axiom 31. m�n or n<m.

Lemma 32. Assume that l < n. Then m+ l <m+n and l+m<n+m.

Lemma 33. Assume that m is nonzero and l <n. Then m � l <m �n and l �m<n �m.

Axiom 34. n=0 or n=1 or 1<n.

Lemma 35. If m=/ 0 then n�n �m.

2.8 Induction

Naproche has inherited an elegant treatment of induction from the SAD system. Naproche
has a special binary relation symbol � for a universal inductive relation: if at any point
m property P is inherited at m provided all �-predecessors of m satisfy P , then P holds
everywhere.

So to prove that P holds universally, it suffices to prove the �inheritance� along �. This
modification of proof tasks is already carried out by the parser when it comes across the
keyword �proof by induction�. This will be demonstrated in a later proof in this Introduc-
tion.

Axiom 36. If n<m then n�m.

From this axiom one can derive Peano axioms for the natural numbers. On the other hand,
with some simple axioms about the successor operation +1 and with � one could have
derived all the above structural axioms.

2.9 Division

We now get to notions that are crucial for the study of divisors and prime numbers:

Definition 37. n divides m iff for some l m=n � l.

Let xjy denote x divides y. Let a divisor of x denote a natural number that divides x.

The definition is similar to the definition of �. Note, however, the possible syntactic
variations: �there exists a natural number l such that m+ l=n�, �for some l m=n � l�. It
is also possible to put the quantifier after the property: �n dividesm iff m=n� l for some l�.

Natural language has many mechanisms for putting information into sentences in a compact,
un-formalistic way. Un-formalistic means, e.g., that natural language does not generally
allow brackets (...) in speech. ForTheL implements several of these natural language mech-
anisms. Although the language has evolved, �The syntax and semantics of the ForTheL
language� by Andrei Paskevich is still a good guide to most constructs of the language.

2.10 An Interactive Proof

We shall later need a technical lemma on divisibility:

Lemma 38. Let l jm and l jm+n. Then l jn.

On the computer I am using,Naproche does not find a proof on its own: depending on some
default timeouts the proof search is abandoned, and the goal l jn fails. In Isabelle/Naproche
this is signaled in the output window, and the failed goal gets an underlining in red.

So the user has to �interactively� supply a proof, which in a first approximation is a list
of statements which leads up to the claim, and which Naproche's ATP is able to prove
successively. Proof statements can also introduce assumptions and new variables to the
argument, and they can structure the proof.

Lemma 39. Let l jm and l jm+n. Then l jn.

Proof. Assume that l is nonzero. Take p such thatm= l� p. Take q such thatm+n= l� q.

Let us show that p� q. Proof by contradiction. Assume the contrary. Then q < p. m+n=
l � q < l � p=m. Contradiction. qed.

Take r= q¡ p. We have (l � p)+ (l � r)= l � q=m+n=(l � p)+n. Hence n= l � r. �

When Naproche encounters a statement immediately followed by an explicit proof then
Naproche defers proving the statement and first goes through the proof. Since proofs may
contain subproofs, this process may take place recursively.

Proofs of a �toplevel� Lemma or Theorem use the

\begin{proof} . . . \end{proof}

environment well-known from LATEX. In our proof there is also a �lowlevel� proof of p� q
indicated by �Let us show that�. Let us discuss some aspects of the proof:

� Most sentences in a proof are statements, or statements extended by certain con-
structs that organize the flow of the argument.

� �Assume that l is nonzero.� is an assumption that introduces the premise �l is
nonzero� to the argument. Instead of �Assume that� one could also use variants like
[let us | we can] (assume | presume | suppose) [that].

� �Take p such that m= l � p.� introduces a new variable p with a specific property to
the argument. To verify this construct the prover has to show the existence of some
object satisfying the property. Again there are variants: [let us | we can] (choose |
take | consider).

� �Let us show that p� q.� claims that the statement p� q holds and announces a
subsequent proof. Alternatives: [let us | we can] (prove | show | demonstrate) (that).

� �Proof by contradiction� denotes the start of an indirect proof. It is recommended
to explicitly mark indirect proof. Note that in the example this is a �lowlevel� proof
that uses a simple

Proof [by ...](.) ... (qed. | end.)

environment instead of the LATEXproof environment.

� Other proof methods are �by cases� and �by induction�.

� �Assume the contrary.�: The contrary is the negation of the current thesis which in
this case is the statement claimed just before. �thesis� denotes the current thesis,
�contradiction� stands for �false�.

� �Then q < p.�: Words like �then�, �hence�, �thus�, �therefore�, �consequently� are
filler words which are redundant for Naproche but may help human readers to
understand the text.

� �m+ n= l � q < l � p=m�: binary relations like �=� or �<� can be chained. The
statement means the conjunction of the single relations. These will be checked from
left to right.

� �Contradiction. qed.�: The indirect proof has reached the desired contradiction, and
that proof environment is closed by �qed.�.

Naproche is able to prove the next lemma without an explicit proof in the text.

Lemma 40. Let mjn=/ 0. Then m�n.

2.11 Primes

Prime numbers are defined as usual. Indeed we define the adjective �prime� which will
enable us to write �prime natural number� or �prime divisor�.

Let x is nontrivial stand for x=/ 0 and x=/ 1.

Definition 41. n is prime iff n is nontrivial and for every divisor m of n m=1 or m=n.

2.12 Proof by Induction

The following lemma obviously holds by induction: either k is prime itself, or k has a
divisor strictly between 1 and k; by induction that divisor has a prime divisor which is also
a prime divisor of k.

Lemma 42. Every nontrivial k has a prime divisor.

Proof. Proof by induction. �

The phrase �proof by induction� invokes a general induction principle for the relation �.
To prove 8k�(k), it suffices to prove:

8v0(8v1 (v1� v0! �(v1))! �(v0):

So �proof by induction� transforms the thesis into a new thesis:

thesis: forall v0 ((aNaturalNumber(v0) and (not v0 = 0 and not v0 = 1))

implies ((InductionHypothesis :: forall v1 ((aNaturalNumber(v1) and

(not v1 = 0 and not v1 = 1)) implies (iLess(v1,v0)

implies exists v2 ((aNaturalNumber(v2) and doDivides(v2,v1))

and isPrime(v2))))) implies exists v1

((aNaturalNumber(v1) and doDivides(v1,v0)) and isPrime(v1))))

Note that internally, iLess represents the relation �. Since we had postulated axiomati-
cally that< is a subrelation of�, the induction principle for� implies a standard induction
principle for the natural numbers.

2.13 Classes

�sets� and �classes� are built-in notions of Naproche Naproche, as well as �element of ...�.
�element of y� determines the type of elements-of-y. Such �of�-types lead to several lin-
guistic modifications: one can quantify over elements-of-y like (for all | for some | no)
(element of y); y has some element; etc.

Similarly, the subclass relation is given by the dependent type of subclasses-of-T .

Let S; T stand for classes. Let x belongs to S stand for x is an element of S.

Definition 43. A subclass of S is a class T such that every element of T belongs to S.

Let T �S stand for T is a subclass of S.

To avoid the classical antinomies of set theory, classes can only have �small� elements which
in Naproche's terminology are �setsized�; both these adjective were identified earlier in
the document. We extend the built-in ontology of Naproche according to the following
principles:

Axiom 44. Every element of every class is small.

Axiom 45. Every set is a small class and every small class is a set.

Classes can be naturally formed in ForTheL:

Definition 46. N is the class of natural numbers.

The verbose form is equivalent to the use of abstraction terms:

Definition 47. fm; :::; ng= f natural number ijm� i�ng.

2.14 Finite Sequences and Products, using Intuitive �...�-Notation

This section demonstrates how some notation that is usually considered vague can be
interpreted as formally exact. Mathematics often uses ...-notations to indicate arbitrary
size finite or even infinite mathematical objects.

From a LATEXstandpoint, a notation like fm;:::;ng can canonically be seen as the printout
of a corresponding macro in the LATEXsource. Naproche on the other hand accepts stan-
dard LATEXmacros as patterns, so that the macro can be a properly introduced Naproche
pattern with a first-order definition. In this way, intuitive and customary notation can be
used also as Naproche input.

In the present text, fm; :::; ng is printed from a macro defined by:

\newcommand{\Seq}[2]{\{#1,\dots,#2\}}

This notation or macro can be given a precise semantics by ForTheL definitions.

Definition 48. fm; :::; ng is the class of natural numbers i such that m� i�n.

So far there are basically no axioms for set formation, so we postulate:

Axiom 49. fm; :::; ng is a set.

The macro for the fm; :::; ng- notation is visible in the LATEXcode:

\begin{definition} $\Seq{m}{n}$ is the class of

natural numbers $i$ such that $m \leq i \leq n$.

\end{definition}

2.15 Functions

The notion of �function� and some related notations like the domain of a function F or
the application F (x) of a function to an argument are provided by Naproche. We add an
axiom about smallness of values:

Axiom 50. Assume F is a function and x2Dom(F ). Then F (x) is small.

Definition 51. A sequence of length n is a function F such that Dom(F )= f1; :::; ng.

The members F (i) of a sequence F are often written in an indexed notation fi where this
notation is defined by a macro

\newcommand{\val}[2]{#1_{#2}}

The ForTheL semantics is defined by:

Let Fi stand for F (i).

Definition 52. Let F be a sequence of length n. fF1; :::; Fng= fFiji2Dom(F )g.

Dot notation is also used for iterations of all sorts. For Euclid's theorem we shall want to
consider products of finitely many prime numbers. So we postulate axiomatically:

Signature 53. Let F be a sequence of length n such that fF1; :::; Fng�N. F1���Fn is a
natural number.

Axiom 54. (Factorproperty) Let F be a sequence of length n such that F (i) is a nonzero
natural number for every i2Dom(F ). Then F1���Fn is nonzero and F (i) divides F1���Fn
for every i2Dom(F ).

Note that we can name toplevel sections by single words like �Factorproperty� or numbers.
These can be referenced later in the form �(by Factorproperty)�.

2.16 Finite and Infinite Sets

Finite sequences readily allow a formalization of finiteness for arbitrary sets and classes.

Definition 55. S is finite iff S=fF1; :::;Fng for some natural number n and some function
F that is a sequence of length n.

Definition 56. S is infinite iff S is not finite.

2.17 Euclid's Theorem

Now everything is in place for the proof that there are infinitely many prime numbers.

Signature 57. P is the class of prime natural numbers.

Theorem 58. (Euclid) P is infinite.

Proof. Assume that r is a natural number and p is a sequence of length r and fp1; :::; prg
is a subclass of P.

(1) pi is a nonzero natural number for every i such that 1� i� r.

Consider n= p1���pr+1. Take a prime divisor q of n.

Let us show that q=/ pi for all i such that 1� i� r.

Proof by contradiction. Assume that q= pi for some natural number i such that 1� i� r.
q is a divisor of n and q is a divisor of p1���pr (by Factorproperty, 1). Thus q divides 1.
Contradiction. qed.

Hence fp1; :::; prg is not the class of prime natural numbers. �

